Termodinámica de procesos irreversibles

Drop fluid inmerse in a viscous fluid of different viscosity

Initialization

```
Needs["VectorAnalysis`"]
```

Radial functions inside and outside the drop

```
fi[r_] := A r^2/4 + B r^4/8
fe[r_] := a r + b/r
U = u {Cos[Ttheta], -Sin[Ttheta], 0};
```

Velocity Field inside and outside

```
V[f_] := Curl[Curl[f U, Spherical], Spherical]
vi = V[fi[Rr]] // Simplify
ve = V[fe[Rr]] + U // Simplify
```

$$\{-(A + B\,Rr^2)\,u\,Cos[Ttheta],\ (A + 2\,B\,Rr^2)\,u\,Sin[Ttheta],\ 0\}$$

$$\left\{\frac{(2\,b + Rr^2\,(-2\,a + Rr))\,u\,Cos[Ttheta]}{Rr^3},\ \frac{(b + (a - Rr)\,Rr^2)\,u\,Sin[Ttheta]}{Rr^3},\ 0\right\}$$

Pressure without gravity force

```
pe = Integrate[η Laplacian[ve, Spherical], Rr][[1]] // Simplify
pe =
 Rr Integrate[η Laplacian[ve, Spherical], Ttheta][[2]] // Simplify
pi = Integrate[ν Laplacian[vi, Spherical], Rr][[1]] // Simplify
pi =
 Rr Integrate[ν Laplacian[vi, Spherical], Ttheta][[2]] // Simplify
```

$$-\frac{2\,a\,u\,\eta\,Cos[Ttheta]}{Rr^2}$$

$$-\frac{2\,a\,u\,\eta\,Cos[Ttheta]}{Rr^2}$$

$$-10\,B\,Rr\,u\,\nu\,Cos[Ttheta]$$

$$-10\,B\,Rr\,u\,\nu\,Cos[Ttheta]$$

Pressure including gravity

```
peT = pe0 0 + ρe g Rr Cos[Ttheta] + pe // Simplify
piT = pi0 0 + ρi g Rr Cos[Ttheta] + pi // Simplify
```

$$\left(-\frac{2\,a\,u\,\eta}{Rr^2} + g\,Rr\,\rho e\right) \text{Cos}[\text{Ttheta}]$$

$$Rr\,(-10\,B\,u\,\nu + g\,\rho i)\,\text{Cos}[\text{Ttheta}]$$

Stress Tensor components

```
SeRr = -peT + 2 η D[ve[[1]], Rr] // Simplify
SiRr = -piT + 2 ν D[vi[[1]], Rr] // Simplify

SeTR = η (D[ve[[1]], Ttheta] / Rr + D[ve[[2]], Rr] - ve[[2]] / Rr) //
  Simplify
SiTR = ν (D[vi[[1]], Ttheta] / Rr + D[vi[[2]], Rr] - vi[[2]] / Rr) //
  Simplify
```

$$-\frac{\left(12\,b\,u\,\eta - 6\,a\,Rr^2\,u\,\eta + g\,Rr^5\,\rho e\right) \text{Cos}[\text{Ttheta}]}{Rr^4}$$

$$Rr\,(6\,B\,u\,\nu - g\,\rho i)\,\text{Cos}[\text{Ttheta}]$$

$$-\frac{6\,b\,u\,\eta\,\text{Sin}[\text{Ttheta}]}{Rr^4}$$

$$3\,B\,Rr\,u\,\nu\,\text{Sin}[\text{Ttheta}]$$

Boundary condition impossed to determine the unknown constants; a,b,A,B, u. Results

```
Sys = {ve[[1]] == 0, vi[[1]] == 0,
   ve[[2]] == vi[[2]], SeTR - SiTR == 0, SeRr - SiRr == 0}
Solve[Sys, {a, b, A, B, u}] // Simplify
```

$$\left\{\frac{\left(2\,b+Rr^2\,(-2\,a+Rr)\right)\,u\,Cos[Ttheta]}{Rr^3}==0,\ -\left(A+B\,Rr^2\right)\,u\,Cos[Ttheta]==0,\right.$$
$$\frac{\left(b+(a-Rr)\,Rr^2\right)\,u\,Sin[Ttheta]}{Rr^3}==\left(A+2\,B\,Rr^2\right)\,u\,Sin[Ttheta],$$
$$-\frac{6\,b\,u\,\eta\,Sin[Ttheta]}{Rr^4}-3\,B\,Rr\,u\,\nu\,Sin[Ttheta]==0,$$
$$\left.-\frac{\left(12\,b\,u\,\eta-6\,a\,Rr^2\,u\,\eta+g\,Rr^5\,\rho e\right)\,Cos[Ttheta]}{Rr^4}-Rr\,(6\,B\,u\,\nu-g\,\rho i)\,Cos[Ttheta]==0\right\}$$

$$\left\{\left\{A\to\frac{\eta}{2\,(\eta+\nu)},\ b\to\frac{Rr^3\,\nu}{4\,(\eta+\nu)},\ a\to\frac{Rr\,(2\,\eta+3\,\nu)}{4\,(\eta+\nu)},\ u\to\frac{2\,g\,Rr^2\,(\eta+\nu)\,(\rho e-\rho i)}{3\,\eta\,(2\,\eta+3\,\nu)},\ B\to-\frac{\eta}{2\,Rr^2\,(\eta+\nu)}\right\}\right\}$$